Adresse aus Outlook

Wenn Sie in Outlook Kontakte erstellt haben, so können Sie diese mit kompletter Anschrift
in ein Word Dokument einfügen. Dies ist vor allen Dingen bei Briefen sehr praktisch. Allerdings
ist diese Funktion nicht über die Menüleiste von Word verfügbar, sondern nur über ein Icon
in der Symbolleiste, was allerdings bei der Standardinstallation von Word nicht automatisch
eingefügt wird.

Um dieses Icon nachträglich hinzuzufügen, gehen Sie wie folgt vor:

1. Klicken Sie in der Menü-Leiste auf Ansicht Symbolleiste Anpassen
2. Klicken Sie auf den Registerreiter Befehle
3. Klicken Sie im linken Fenster auf den Eintrag "Einfügen" und blättern Sie im rechten Fenster so weit herunter, bis das Icon "Adressbuch" kommt.
4. Gehen Sie mit der Maus auf das das Adressbuch-Icon, halten Sie die linke Maustaste gedrückt, ziehen Sie das Icon auf einen beliebigen Platz in der Symbolleiste und lassen die linke Maustaste los.
5. Beenden Sie das Fenster "Anpassen" mit einem Klick auf Schließen

Wenn Sie jetzt auf das neue Adressbuch-Icon klicken, zeigt Ihnen Word Ihre eingegebenen Kontakte
an, die Sie mit einem Doppelklick auf den Namen einfügen können.

Zusatztipp

Sollte es beim Einfügen einer Adresse vorkommen, daß Word nur den Namen, nicht aber den Rest der Anschrift übernimmt, so liegt das daran, daß Sie die Adresse des Kontaktes in Outlook unter dem Eintrag "Privat" eingegeben haben. Wenn Sie die Adresse Ihres Kontaktes jedoch unter "Geschäftlich"
abspeichern, sollte es ordnungsgemäß funktionieren.

So können Sie per VBA auf das Adressbuch von Outlook zugreifen
Versionen: Word 2003, 2002/XP und 2000

Wenn Sie aus Ihren VBA-Makros auf das Outlook-Adressbuch zugreifen möchten, stoßen Sie schnell auf ein Problem: Für den Zugriff auf das Adressbuch stellt das Outlook-Objektmodell keine Möglichkeiten bereit. Um eine Adressauswahl zu realisieren ist deshalb ein Umweg über CDO (Collaboration Data Objects) notwendig. Dort steht eine Funktion "AddressBook" zur Verfügung, die als Ergebnis eine "Recipients"-Auflistung mit den gewählten Empfängern liefert. Voraussetzung für die Nutzung dieser Funktion ist zunächst, dass CDO auf dem System installiert ist. Prüfen können Sie das, indem Sie die VBA-Entwicklungsumgebung öffnen, dort das Menü Extras-Verweise anwählen und nach einem Eintrag "Microsoft CDO 1.2x Object Library" suchen. Ist der Eintrag nicht vorhanden, gehen Sie wie folgt vor, um CDO nachträglich zu installieren:

1. Beenden Sie Outlook und alle anderen, eventuell laufenden Office-Anwendungen.

2. Legen Sie die Office-CD ein und starten Sie dort "Setup".

3. Klicken Sie auf die Schaltfläche Features Hinzufügen/entfernen.

4. Wenn Sie Office 2003 einsetzen, aktivieren Sie die Option Erweiterte Anpassung Von Anwendungen und klicken Sie dann auf Weiter.

5. Aktivieren Sie im Bereich "Microsoft Outlook für Windows" für den Eintrag "Collaboration Data Objects" die Option Vom Arbeitsplatz Starten.

6. Klicken Sie auf die Schaltfläche Jetzt Aktualisieren und verlassen Sie das Setup anschließend wieder.

Danach können Sie Outlook wieder starten, in der VBA-Entwicklungsumgebung ist nun unter Extras-Verweise der Eintrag "Microsoft CDO 1.2x Object Library" vorhanden und kann über das Kontrollkästchen aktiviert werden.

Um das Adressbuch für die Auswahl einer oder mehrere Adressen anzeigen zu lassen, geben Sie folgende Deklarationen und die Funktion "AddrBookTest()" in ein vorhandenes oder in ein neues Modul ein:

Option Explicit

Const txtTitel = "Word Weekly - AddrBookTest"

Function AddrBookTest(OneAddress As Boolean)
  Dim objCDO As MAPI.Session
  Dim objRecs As MAPI.Recipients
  Dim strResult As String, strErr As String, I&

  On Error Resume Next
  AddrBookTest = "" 'Default: Nichts ausgewählt
  Set objCDO = CreateObject("MAPI.Session")
  objCDO.Logon "", "", False, False
  If Err <> 0 Or objCDO Is Nothing Then
    Beep
    MsgBox "Fehler bei Initialisierung " & _
           "der CDO-Schnittstelle: " _
           & Err.Description, _
           vbOKOnly + vbExclamation, txtTitel
    GoTo Exit_Func
  End If

  With objCDO.Session
    Err = 0
    Set objRecs = .AddressBook(Title:="Empfänger wählen:")
    DoEvents
    If Err <> 0 Then
      strErr = Err.Description
      If InStr(LCase$(strErr), "user_cancel") = 0 Then
        Beep
        MsgBox "Problem beim Aufruf des " & _ 
               "Adressbuches (" & strErr & ")...", _
               vbOKOnly + vbExclamation, txtTitel
        GoTo Exit_Func
      End If 'User_Cancel?
    End If 'Err<> 0...
  End With
  If objRecs Is Nothing Then Exit Function

  Err = 0
  strResult = ""
  For I = 1 To objRecs.Count
    With objRecs.Item(I)
      strResult = strResult & _
                  Trim$(.AddressEntry.Name) & _
      "|" & Trim$(.AddressEntry.Address) & vbCrLf
    End With
    If OneAddress And I = 1 Then Exit For
  Next I

  If Err <> 0 Then
    strErr = Err.Description
    Beep
    MsgBox "Problem bei der Übernahme der Adresse (" & _
           strErr & ")...", _
           vbOKOnly + vbExclamation, txtTitel
  Else 'letztes vbCrLf wieder raus
    strResult = Left$(strResult, Len(strResult) - 2)
    AddrBookTest = strResult
  End If

Exit_Func:
  objCDO.Logoff

  Set objRecs = Nothing
  Set objCDO = Nothing

End Function

Diese Funktion initialisiert zunächst eine CDO-Session mit dem Standard-Profil, fängt dabei einen eventuell auftretenden Fehler ab und gibt diesen gegebenenfalls zur Kenntnis. Über diese Session kann dann die Funktion "AddressBook" aufgerufen werden, die als Ergebnis eine "Recipients"-Auflistung liefert. Eventuell auftretende Fehler untersuchen wir auf die Zeichenfolge "user_cancel", die einen Klick auf Abbrechen signalisiert. Alle anderen Fehler werden entsprechend per Meldung zur Kenntnis gegeben, bevor die Routine verlassen wird.

Die Eigenschaft "Count" der "Recipients"-Auflistung gibt Auskunft, wie viele Einträge ausgewählt wurden. Dementsprechend gehen wir in einer Schleife alle Einträge durch und setzen diese zu einem Funktionsergebnis zusammen, das zeilenweise Namen und Adressen durch das Pipe-Zeichen "|" getrennt enthält. Wenn der Parameter "OneAddress" mit "True" angegeben war, verlassen wir die Schleife nach Bearbeitung des ersten Eintrages. Einen eventuell auftretenden Fehler, beispielsweise ausgelöst durch einen Klick auf Nein in der bei Zugriffen auf E-Mail-Adressen angezeigten Sicherheitsabfrage von Outlook, fangen wir entsprechend ab und zeigen diesen per Meldung an.

Das Funktionsergebnis ist entweder eine Zeichenfolge wie zum Beispiel "Vorname Nachname|nachname@provider.de" (OneAddress= True), also eine Kombination aus dem Namen und der E-Mail-Adresse oder eine per "vbCrLf" getrennte Liste oben genannten Aufbaus (OneAddress= False). Um auf die einzelnen Einträge der Liste zuzugreifen, wandeln Sie sie zunächst per "Split" in ein Array um und zerlegen Sie dann anhand des Trenners "|" in Name und E-Mail-Adresse:

Sub Test_1a()
  Dim strListe As String, X As Variant, I&
  Dim strName As String, strEMail As String

  strListe = AddrBookTest(False)
  X = Split(strListe, vbCrLf)
  For I = 0 To UBound(X)
    strName = Left$(X(I), InStr(X(I), "|") - 1)
    strEMail = Mid$(X(I), InStr(X(I), "|") + 1)
    Debug.Print strName, strEMail
  Next I

End Sub

Eine einzelne Adresse zerlegen Sie wie oben gezeigt per "Left$()" und "Mid$()" in Ihre Bestandteile:

Sub Test_1b()
  Dim strAusw As String, strName As String, _
      strEMail As String

  strAusw = AddrBookTest(True)
  strName = Left$(strAusw, InStr(strAusw, "|") - 1)
  strEMail = Mid$(strAusw, InStr(strAusw, "|") + 1)
  Debug.Print strName, strEMail

End Sub

